

# Propositional Calculus in Coq

Floris van Doorn

May 9, 2014

## Abstract

I formalize important theorems about classical propositional logic in the proof assistant Coq. The main theorems I prove are (1) the soundness and completeness of natural deduction calculus, (2) the equivalence between natural deduction calculus, Hilbert systems and sequent calculus and (3) cut elimination for sequent calculus.

## 1 Introduction

Proof assistants (or interactive theorem provers) are computer programs which help to formalize and check the correctness of proofs. Proof assistants are used for two things. First, they are used for formal verification of hardware or software, which means proving that the hardware or software conforms to the specifications. Second, they are used to formally verify mathematical theorems. I will only discuss the second use in this paper.

To prove a mathematical theorem in a proof assistant, the user has to write down the proof in a language that the proof assistant can parse, and then the proof assistant will check correctness. Usually the proof assistant provides small amounts of automation, so that the user does not have to write down every detail. Still, a user typically has to give a lot more details to a proof assistant than is usually given on a paper proof. This means that a formalization of a mathematical theorem is much more time consuming than proving the theorem on paper, hence most research mathematics is not formalized. Still, there are some deep theorems which have been fully formalized in proof assistants, such as the four color theorem<sup>[2]</sup> or Feit-Thompson's odd order theorem<sup>[3]</sup>, which were both formalised by Georges Gonthier in the proof assistant Coq. It is very rare that a new result is immediately accompanied by a formalization in a proof assistant, the author knows only two instances where this has been done.<sup>[6,7]</sup>

The goal of this project is to formalize some theorems in proof theory in a proof assistant. I chose to use the proof assistant Coq<sup>[4]</sup> for this. Coq is a proof assistant based on the predicative Calculus of Inductive Constructions, which uses the expressive power of dependent type theory as language. It uses constructive reasoning, which is very suitable for the meta-logic of proof theory, since most proofs can be given constructively.

I have formalized the following three theorems.

1. The completeness theorem of classical natural deduction calculus.
2. The equivalence between the classical versions of natural deduction calculus (Nc), Hilbert-style deduction calculus (Hc) and Gentzen's sequent calculus (Gc).

### 3. Cut elimination for Gentzen's sequent calculus.

To simplify, I only proved this for propositional calculus. Formalizing full predicate calculus gives rise to additional difficulties such as variable encodings, variable renaming, alpha-equivalence and induction over type families such as vectors.

The following table is a summary of the formalization. The number of lines in each file is rounded.

file	Defines	Proves	# lines
a_base	Variable	–	20
b_soundness	Provability in Nc, validity	Soundness of Nc.	150
c_completeness	Conjunctive Normal Form	Completeness of Nc.	400
d_hilbert_calculus	Provability in Hc	Equivalence of Nc and Hc.	100
e_sequent_calculus	Provability in Gc	Equivalence of Nc and Gc.	200
f_cut_elimination	Cut-free sequent calculus	Cut elimination.	180

In total there are just over 1000 lines, 66 theorems and 43 definitions.

In the next sections I will discuss the formalization in more detail. I will focus on the considerations which went into the formalization. I will skip all proofs and most Lemmas, but I will give informal proof sketches for the main theorems. To view all Lemmas, check the coqdoc html files. To view all proofs, check the Coq source files.

## 2 Basic Definitions

In `a_base` I do some preparatory work, by defining variables. The relevant lines are

```
Parameter PropVars : Set.
Hypothesis Varseq_dec : ∀ x y:PropVars, {x = y} + {x ≠ y}.
```

The first line indicates that the set of (propositional) variables is some set. The second line indicates that equality is decidable on this set. Note that since Coq is based on constructive logic, so this is not vacuous. This hypothesis is necessary for the completeness theorem. If we did not assume this, then one could take for example the constructive reals as variables. It is consistent with the constructive reals that all valuations (functions from variables to bool) are constant.<sup>[1]</sup> This means that it is possible that  $\#0 \vee \#1$  (we denote variables with '#') is valid, but it is not provable. So in this case completeness would fail. These are the only axioms or parameters in the formalization. All results work even if there are only finitely many variables, even if there is just one variable, or even none at all (though that wouldn't be a very interesting case).

We can now define formulas in `b_soundness`.

```
Inductive PropF : Set :=
| Var : PropVars → PropF
| Bot : PropF
| Conj : PropF → PropF → PropF
| Disj : PropF → PropF → PropF
| Impl : PropF → PropF → PropF
.
```

Notation "# P" := (Var P) (at level 1) : My\_scope.

Notation "A ∨ B" := (Disj A B) (at level 15, right associativity) : My\_scope.

Notation "A ∧ B" := (Conj A B) (at level 15, right associativity) : My\_scope.

Notation "A → B" := (Impl A B) (at level 16, right associativity) : My\_scope.

**Notation** " $\perp$ " := Bot (at level 0) : *My\_scope*.  
**Definition** Neg  $A := A \multimap \perp$ .  
**Notation** " $\neg A$ " := (Neg  $A$ ) (at level 5) : *My\_scope*.  
**Definition** Top :=  $\neg \perp$ .  
**Notation** " $\top$ " := Top (at level 0) : *My\_scope*.

In the formalization I used the unicode character  $\rightarrow$  for implication, but to distinguish it from the function type of Coq, I use  $\multimap$  in this document instead.

I use defined negation, because that simplifies induction proofs by having one fewer connective. Other than this, I take all connectives as primitive. I could have defined conjunction and disjunction in terms of implication and falsum and retain an equivalent system, since I only formalize classical logic. However, I decided against this, to make the formalization easier to adapt to intuitionistic or minimal logic, where these connectives aren't interdefinable. We defined negation because it is definable in all these calculi.

In an earlier version of the formalization I used a special variable  $\perp$  which corresponds to the false formula. This had as advantages that the variables corresponded exactly to the atomic formulae and that there is one fewer induction base case when proving things by induction over formulae. In the end I decided against it, since it is more natural to have  $\perp$  not as a variable, but as a separate constant, and the alternative required to define valuations as assigning 'false' to the variable  $\perp$ , while no such requirement is necessary now.

A *valuation* is an element of the space  $PropVars \rightarrow \mathbf{bool}$ . I define the truth of a formula  $A$  under valuation  $v$  in the obvious way. This allows us to define the validity of formulae. The map `ls_true` sends a boolean value to the corresponding (true or false) proposition. `[]` denotes the empty list.

```

Fixpoint TrueQ  $v A : \mathbf{bool} := \text{match } A \text{ with}$ 
```

- | #  $P \Rightarrow v P$
- |  $\perp \Rightarrow \text{false}$
- |  $B \vee C \Rightarrow (\text{TrueQ } v B) \mid\mid (\text{TrueQ } v C)$
- |  $B \wedge C \Rightarrow (\text{TrueQ } v B) \ \&\& \ (\text{TrueQ } v C)$
- |  $B \multimap C \Rightarrow (\text{negb } (\text{TrueQ } v B)) \mid\mid (\text{TrueQ } v C)$

```

end.
Definition Satisfies  $v \Gamma := \forall A, \text{In } A \Gamma \rightarrow \text{ls\_true } (\text{TrueQ } v A)$ .
Definition Models  $\Gamma A := \forall v, \text{Satisfies } v \Gamma \rightarrow \text{ls\_true } (\text{TrueQ } v A)$ .
Notation " $\Gamma \models A$ " := (Models  $\Gamma A$ ) (at level 80).
Definition Valid  $A := [] \models A$ .
  
```

The notion of provability is naturally defined inductively. I use a context sharing version of classical natural deduction. Using the context sharing version simplifies proofs, because changes in the context are usually hard to deal with for proofs of meta-theoretic theorems. Note that I am using lists, not sets, for the context, because it is easier to reason about lists in proofs.

```

Reserved Notation " $\Gamma \vdash A$ " (at level 80).
Inductive Nc : list PropF  $\rightarrow$  PropF  $\rightarrow$  Prop :=
| Nax :  $\forall \Gamma A, \text{In } A \Gamma \rightarrow \Gamma \vdash A$ 
| Impl :  $\forall \Gamma A B, A :: \Gamma \vdash B \rightarrow \Gamma \vdash A \multimap B$ 
| ImpE :  $\forall \Gamma A B, \Gamma \vdash A \multimap B \rightarrow \Gamma \vdash A \rightarrow \Gamma \vdash B$ 
| BotC :  $\forall \Gamma A, \neg A :: \Gamma \vdash \perp \rightarrow \Gamma \vdash A$ 
| AndI :  $\forall \Gamma A B, \Gamma \vdash A \rightarrow \Gamma \vdash B \rightarrow \Gamma \vdash A \wedge B$ 
| AndE1 :  $\forall \Gamma A B, \Gamma \vdash A \wedge B \rightarrow \Gamma \vdash A$ 
| AndE2 :  $\forall \Gamma A B, \Gamma \vdash A \wedge B \rightarrow \Gamma \vdash B$ 
| OrI1 :  $\forall \Gamma A B, \Gamma \vdash A \rightarrow \Gamma \vdash A \vee B$ 
| OrI2 :  $\forall \Gamma A B, \Gamma \vdash B \rightarrow \Gamma \vdash A \vee B$ 
  
```

| OrE :  $\forall \Gamma A B C, \Gamma \vdash A \vee B \rightarrow A :: \Gamma \vdash C \rightarrow B :: \Gamma \vdash C \rightarrow \Gamma \vdash C$   
 where "Γ ⊢ A" := (Nc Γ A) : My\_scope.

Definition Provable A := [] ⊢ A.

This allows us to define the propositions we're aiming to prove.

Definition Prop\_Soundness :=  $\forall A, \text{Provable } A \rightarrow \text{Valid } A$ .

Definition Prop\_Completeness :=  $\forall A, \text{Valid } A \rightarrow \text{Provable } A$ .

Given those definitions, Soundness can be proved directly by induction on the derivation (no additional lemmas are needed). Indeed, in the formalization this only requires 17 lines.<sup>1</sup>

Theorem Soundness : Prop\_Soundness.

### 3 Completeness

The formalization continues to prove Completeness. I will first give a proof sketch, before I turn my attention to the formalization.

#### 3.1 Proof Sketch

The formalization uses the following proof. Recall that a literal is an atomic formula or the negation of an atomic formula, a clause is a disjunction of literals, and that a formula is in conjunctive normal form (CNF) if it's a conjunction of clauses. Also, a formula is in *negation normal form* (NNF), if it only consists of conjunctions, disjunctions and literals. In this paper I will use CNF and NNF as nouns indicating formulae in CNF resp. NNF.

*Proof sketch of Completeness.* Let  $A$  be a formula,  $A_{\text{NNF}}$  its negation normal form and  $A_{\text{CNF}}$  its conjunctive normal form. We say that a clause is *syntactically invalid* if either it contains both  $p$  and  $\neg p$  for some atomic formula  $p$ , or if it contains  $\neg \perp$ . We say that  $A_{\text{CNF}}$  is *syntactically valid* if it contains no syntactically invalid clauses. Then Completeness follows from the following statements

- (1) If  $A$  is valid, then  $A_{\text{NNF}}$  is valid.
- (2) If  $A_{\text{NNF}}$  is valid, then  $A_{\text{CNF}}$  is valid.
- (3) If  $A_{\text{CNF}}$  is valid then  $A_{\text{CNF}}$  is syntactically valid.
- (4) If  $A_{\text{CNF}}$  is syntactically valid then  $A_{\text{CNF}}$  is provable.
- (5)  $A_{\text{CNF}} \rightarrow A_{\text{NNF}}$  is provable.
- (6)  $A_{\text{NNF}} \rightarrow A$  is provable.

All statements except (3) are proven by induction to the structure of the formula in question. Many of these proofs need additional lemmas, which are also proven by induction. Statement (3) is most naturally proved using contraposition, by showing that if  $A_{\text{CNF}}$  is syntactically invalid, then there is a countervaluation  $v$  for it. First one does this for a clause. The countervaluation  $v$  of a syntactically invalid clause  $C$  can be defined as  $v(x) = \text{true}$  iff  $\neg x$  occurs in  $C$ . One can then

---

<sup>1</sup>Counting the lines for theorems Soundness\_general and Soundness and for tactics case\_bool and prove\_satisfaction. No other lemmas or tactics are used.

show that  $v$  is indeed a valuation, and that  $v$  makes  $C$  false. Then if  $A_{\text{CNF}}$  is syntactically invalid, it contains some syntactically invalid clause  $C$ , and the countervaluation for that clause is also a countervaluation for  $A_{\text{CNF}}$ .  $\square$

### 3.2 Considerations for the formalization

In the formalization I defined clauses as lists of literals and CNFs (formulae in CNF) as lists of clauses. The alternative, to say that only some class of propositional formulae are in CNF, will be very hard to work with. One of the problems which one has to deal with is associativity. Which of the following do you call clauses?

$$(((x \vee \neg y) \vee z) \vee w) \quad (x \vee (\neg y \vee (z \vee w))) \quad (x \vee \neg y) \vee (z \vee w).$$

You can call all of them clauses, but then clauses are not in a canonical form, and proving theorems about them will probably be harder. Also, induction proofs will be harder, because you're constantly proving or using the fact that some formula is in the shape of a CNF. Using lists, you can make sure that all clauses have the same shape of parentheses, and a term of the correct type is automatically in CNF. The disadvantage, that you have to define a mapping from clauses to actual formulas, is very minor.

A viable alternative to using lists would be using non-empty lists, i.e. lists which have at least one element. When transforming a formula to conjunctive normal form, it turns out we never end up with an empty clause (i.e. an empty list of literals) or an empty formula in CNF (i.e. an empty list of clauses). Using non-empty lists might be more natural in that case, but I decided against this, because I thought this would make meta-theorems harder to prove. With non-empty lists, the base case is a list of one element, and that case is probably harder than the case of an empty list. However, allowing for empty lists gives an unexpected artefact in the translation from lists to formulae. When transforming a clause (a list of literals) to a formulae, it is natural to say  $f(\text{nil}) = \perp$  (nil is the empty list) and  $f(a::l) = a \vee f(l)$ . However, this means that, for example,  $f([x; y; z]) = x \vee (y \vee (z \vee \perp))$  instead of the more natural  $f([x; y; z]) = x \vee (y \vee z)$ . This poses no problems except that it is unnatural.

### 3.3 Definitions

The definitions needed for Completeness are given below, given in `c_completeness`.

```

Inductive NNF : Set :=
| NPos : PropVars → NNF
| NNeg : PropVars → NNF
| NBot : NNF
| NTop : NNF
| NConj : NNF → NNF → NNF
| NDisj : NNF → NNF → NNF
Fixpoint MakeNNF (A:PropF) : NNF := match A with
| # P ⇒ NPos P
| ⊥ ⇒ NBot
| B ∨ C ⇒ NDisj (MakeNNF B) (MakeNNF C)
| B ∧ C ⇒ NConj (MakeNNF B) (MakeNNF C)
| B → C ⇒ NDisj (MakeNNFN B) (MakeNNF C)

```

```

end
with MakeNNFN (A:PropF) : NNF := match A with
| # P ⇒ NNeg P
| ⊥ ⇒ NTop
| B ∨ C ⇒ NConj (MakeNNFN B) (MakeNNFN C)
| B ∧ C ⇒ NDisj (MakeNNFN B) (MakeNNFN C)
| B → C ⇒ NConj (MakeNNFN B) (MakeNNFN C)
end.
Inductive Literal :=
| LPos : PropVars → Literal
| LNeg : PropVars → Literal
| LBot : Literal
| LTop : Literal
.

```

The inclusions `NNFtoPropF` and `LiteraltoPropF` from respectively NNF and literals to propositional formulae are defined in the obvious way.

In `a_base` I define the constructor `map_fold_right`, such that for  $f : B \rightarrow A$ ,  $g : A \rightarrow A \rightarrow A$ ,  $a : A$  and  $x_i : B$  we have

$$\text{map\_fold\_right } f \ g \ a \ [x_0; \dots; x_n] = g(f(x_0), g(f(x_1), \dots, g(f(x_n), a) \dots)).$$

The definition is below, and allows us to define the inclusion from Clause and CNF to PropF.

```

Fixpoint map_fold_right (A B:Type) (f : B → A) (g : A → A → A) a l := match l with
| nil ⇒ a
| b::l2 ⇒ g (f b) (map_fold_right f g a l2)
end.
Definition Clause := list Literal.
Definition ClausetoPropF := map_fold_right LiteraltoPropF Disj ⊥.
Definition CNF := list Clause.
Definition CNFtoPropF := map_fold_right ClausetoPropF Conj ⊤.

```

We still need to define the map which transforms a NNF to a CNF. The conjunction of two CNFs is just concatenation of the corresponding lists. However, the disjunction is harder to define. Doing first the simpler case of taking the disjunction of a clause with a CNF, we see that this corresponds to just adding this clause in front of every other clause:

$$C \vee (C_1 \wedge C_2 \wedge \dots) \rightsquigarrow (C \vee C_1) \wedge (C \vee C_2) \wedge \dots,$$

see `AddClause` below (`map f l` means applying  $f$  to every element of  $l$ ). We can then define the disjunction of two CNFs as follows:

$$(C_1 \wedge C_2 \wedge \dots) \vee A \rightsquigarrow (C_1 \vee A) \wedge (C_2 \vee A) \wedge \dots$$

where  $A$  is an CNF, and where  $C_1 \vee A$  is defined using `AddClause`. This gives rise to the definition of `Disjunct` below. The term `flat_map` is defined in the Coq library such that if  $f : A \rightarrow \text{list } B$  and  $x_i : A$  then

$$\text{flat\_map } f \ [x_0; \dots; x_n] = f(x_0) ++ f(x_1) ++ \dots ++ f(x_n).$$

This allows us to define the transformation from NNF to CNF.

```

Definition AddClause (l:Clause) (ll:CNF) : CNF := map (fun l2 => l++l2) ll.
Definition Disjunct (ll ll2:CNF) : CNF := flat_map (fun l => AddClause l ll2) ll.

Fixpoint MakeCNF (A:NNF) : CNF := match A with
| NPos P => [[LPos P]]
| NNeg P => [[LNeg P]]
| NBot => [[LBot]]
| NTop => [[LTop]]
| NConj B C => MakeCNF B ++ MakeCNF C
| NDisj B C => Disjunct (MakeCNF B) (MakeCNF C)
end.

```

Finally, we can define the syntactical validity.

```

Definition Valid-Clause (l:Clause) := In LTop l ∨ ∃ A, (In (LPos A) l ∧ In (LNeg A) l).
Definition Valid-CNF ll := ∀ l, In l ll → Valid-Clause l.

```

### 3.4 Formalized Proof

We now turn to prove the 6 statements described in Section 3.1.

```

Lemma NNF_equiv_valid : ∀ v A, TrueQ v (NNFtoPropF (MakeNNF A)) = TrueQ v A ∧
  TrueQ v (NNFtoPropF (MakeNNFN A)) = TrueQ v ¬A.
Theorem CNF_equiv_valid : ∀ v A, TrueQ v (CNFtoPropF (MakeCNF A)) = TrueQ v (NNFtoPropF A).
Theorem CNF_valid : ∀ ll, Valid (CNFtoPropF ll) → Valid-CNF ll.
Theorem CNF_provable : ∀ ll, Valid-CNF ll → Provable (CNFtoPropF ll).
Theorem CNF_impl_prov : ∀ A, Provable (CNFtoPropF (MakeCNF A)) → NNFtoPropF A.
Lemma NNF_impl_prov : ∀ A, Provable (NNFtoPropF (MakeNNF A)) → A) ∧
  Provable (NNFtoPropF (MakeNNFN A)) → ¬A).

```

In a previous version I proved `CNF_valid` with contraposition as described above. In that version I needed to first prove that syntactical validity was decidable. In the latest version I prove this directly, by showing that a clause is valid if it is valid under the countervaluation for it (which assigns true to a variable  $p$  iff  $\neg p$  occurs in the clause).

This allows us to prove completeness:

```

Theorem Completeness : Prop_Completeness.

```

## 4 Equivalence between the calculi

We now turn to formalizing the equivalence between the natural deduction calculus `Nc`, Hilbert-style calculus `Hc` and Gentzen's sequent calculus `Gc`.

### 4.1 Nc and Hc

In `d_hilbert_calculus` I define Hilbert-style calculus. Natural deduction calculus and Hilbert-style calculus are quite similar. The difference is that in Hilbert-style calculus we have axioms, which allows us to only have implication elimination (modus ponens) as inference rule. The Hilbert-style calculus is defined below.

```

Inductive AxiomH : PropF → Prop :=
| HOr1 : ∀ A B, AxiomH (A → A ∨ B)
| HOr2 : ∀ A B, AxiomH (B → A ∨ B)

```

```

| HAndI :  $\forall A B, \mathbf{AxiomH} (A \rightarrow B \rightarrow A \wedge B)$ 
| HOrE :  $\forall A B C, \mathbf{AxiomH} (A \vee B \rightarrow (A \rightarrow C) \rightarrow (B \rightarrow C) \rightarrow C)$ 
| HAndE1 :  $\forall A B, \mathbf{AxiomH} (A \wedge B \rightarrow A)$ 
| HAndE2 :  $\forall A B, \mathbf{AxiomH} (A \wedge B \rightarrow B)$ 
| HS :  $\forall A B C, \mathbf{AxiomH} ((A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C)$ 
| HK :  $\forall A B, \mathbf{AxiomH} (A \rightarrow B \rightarrow A)$ 
| HClas :  $\forall A, \mathbf{AxiomH} (\neg(\neg A) \rightarrow A)$ 

```

```

Inductive Hc : list PropF  $\rightarrow$  PropF  $\rightarrow$  Prop :=
| Hass :  $\forall A \Gamma, \text{In } A \Gamma \rightarrow \Gamma \vdash_{\mathbf{H}} A$ 
| Hax :  $\forall A \Gamma, \mathbf{AxiomH} A \rightarrow \Gamma \vdash_{\mathbf{H}} A$ 
| HImpE :  $\forall \Gamma A B, \Gamma \vdash_{\mathbf{H}} A \rightarrow B \rightarrow \Gamma \vdash_{\mathbf{H}} A \rightarrow \Gamma \vdash_{\mathbf{H}} B$ 
where "  $\Gamma \vdash_{\mathbf{H}} A$  " := (Hc  $\Gamma A$ ) : My_scope.

```

I then prove the equivalence of these systems.

**Theorem** Nc\_equiv\_Hc :  $\forall \Gamma A, \Gamma \vdash A \leftrightarrow \Gamma \vdash_{\mathbf{H}} A$ .

The proof is not hard. In the direction from left to right, one basically needs to prove that all axioms for Hc are provable in Nc. In the converse direction, one needs to prove the Deduction Theorem for Hc, which states that

$$\Gamma, A \vdash_{\mathbf{Hc}} B \iff \Gamma \vdash_{\mathbf{Hc}} A \rightarrow B.$$

## 4.2 Nc and Gc

In `e_sequent_calculus` I define Gentzen's sequent calculus. Gentzen's sequent calculus is quite different than natural deduction. In sequent calculus the propositions are *sequents* of the form  $\Gamma \vdash \Delta$ , where  $\Gamma$  and  $\Delta$  are either sequences, multisets or sets of formulae. A sequent intuitively means that the conjunction of  $\Gamma$  implies the disjunction of  $\Delta$ . Instead of having introduction and elimination rules as in natural deduction, there are left rules and right rules for each connective, which introduce that connective on that side of the sequent.

I define the sequent calculus in the formalization below. Since lists are easiest to work with, I use them in sequents. Usually when lists are used in sequent calculus, there are some structural rules allowing to move formulas around in the list. However, having these rules make induction proofs very hard, and requires more nonstructural induction proofs, so instead I defined the sequent calculus in such a way that it is possible to apply a rule anywhere in the lists (ordinarily the relevant formulas appear only on one end of the lists). I use the symbol  $\supset$  as separator in a sequent.

```

Inductive G : list PropF  $\rightarrow$  list PropF  $\rightarrow$  Prop :=
| Gax :  $\forall A \Gamma \Delta, \text{In } A \Gamma \rightarrow \text{In } A \Delta \rightarrow \Gamma \supset \Delta$ 
| GBot :  $\forall \Gamma \Delta, \text{In } \perp \Gamma \rightarrow \Gamma \supset \Delta$ 
| AndL :  $\forall A B \Gamma1 \Gamma2 \Delta, \Gamma1 ++ A :: B :: \Gamma2 \supset \Delta \rightarrow \Gamma1 ++ A \wedge B :: \Gamma2 \supset \Delta$ 
| AndR :  $\forall A B \Gamma \Delta1 \Delta2, \Gamma \supset \Delta1 ++ A :: \Delta2 \rightarrow \Gamma \supset \Delta1 ++ B :: \Delta2 \rightarrow \Gamma \supset \Delta1 ++ A \wedge B :: \Delta2$ 
| OrL :  $\forall A B \Gamma1 \Gamma2 \Delta, \Gamma1 ++ A :: \Gamma2 \supset \Delta \rightarrow \Gamma1 ++ B :: \Gamma2 \supset \Delta \rightarrow \Gamma1 ++ A \vee B :: \Gamma2 \supset \Delta$ 
| OrR :  $\forall A B \Gamma \Delta1 \Delta2, \Gamma \supset \Delta1 ++ A :: B :: \Delta2 \rightarrow \Gamma \supset \Delta1 ++ A \vee B :: \Delta2$ 
| ImpL :  $\forall A B \Gamma1 \Gamma2 \Delta, \Gamma1 ++ B :: \Gamma2 \supset \Delta \rightarrow \Gamma1 ++ \Gamma2 \supset A :: \Delta \rightarrow \Gamma1 ++ A \rightarrow B :: \Gamma2 \supset \Delta$ 
| ImpR :  $\forall A B \Gamma \Delta1 \Delta2, A :: \Gamma \supset \Delta1 ++ B :: \Delta2 \rightarrow \Gamma \supset \Delta1 ++ A \rightarrow B :: \Delta2$ 
| Cut :  $\forall A \Gamma \Delta, \Gamma \supset A :: \Delta \rightarrow A :: \Gamma \supset \Delta \rightarrow \Gamma \supset \Delta$ 
where "  $\Gamma \supset \Delta$  " := (G  $\Gamma \Delta$ ) : My_scope.

```

We now turn to proving the equivalence between Nc and Gc. Neither direction is easy.



In the direction from Gc to Nc we prove that

$$\Gamma \supset \Delta \implies \Gamma \vdash \bigvee \Delta,$$

where  $\bigvee \Delta$  is the disjunction of the formulae in  $\Delta$  (similar to a clause, except that the entries can now be any formulae instead of only literals). However, proving this directly is very hard, since in the definition of sequents we make changes in the middle of the disjunction. This means that in every step we have to use multiple or-eliminations and then or-introductions, which gets really ugly. Instead, we prove the following intermediate lemma:

$$\Gamma \supset \Delta \implies \Gamma, \neg\Delta \vdash \perp,$$

where  $\neg\Delta$  is the element-wise negation of formulae in  $\Delta$ . In the formalization this looks as follows.

**Definition** BigOr := fold\_right Disj  $\perp$ .  
**Notation** " $\bigvee \Delta$ " := (BigOr  $\Delta$ ) (at level 19).  
**Notation** " $\neg\Gamma$ " := (map Neg  $\Gamma$ ) (at level 40).  
**Lemma** G\_to\_Nc\_Neg :  $\forall \Gamma \Delta, \Gamma \supset \Delta \rightarrow \Gamma ++ \neg\Delta \vdash \perp$ .  
**Theorem** G\_to\_Nc :  $\forall \Gamma \Delta, \Gamma \supset \Delta \rightarrow \Gamma \vdash \bigvee \Delta$ .

In the direction from Gc to Nc the major difficulty is proving weakening for Gc. On paper proving weakening is not very hard, but in the formalization the fact that we can apply rules anywhere in the list, and the fact that the weakening also occurs anywhere in the list, makes it quite hard to prove this (for example, we need to distinguish cases in every induction step which of the two formulae occurs earlier in the list).

**Lemma** WeakL :  $\forall \Gamma1 \Gamma2 \Delta A, \Gamma1 ++ \Gamma2 \supset \Delta \rightarrow \Gamma1 ++ A :: \Gamma2 \supset \Delta$ .  
**Lemma** WeakR :  $\forall \Gamma \Delta1 \Delta2 A, \Gamma \supset \Delta1 ++ \Delta2 \rightarrow \Gamma \supset \Delta1 ++ A :: \Delta2$ .  
**Theorem** Nc\_to\_G :  $\forall \Gamma A, \Gamma \vdash A \rightarrow \Gamma \supset [A]$ .

## 5 Cut Elimination

In `f_cut_elimination` I prove the cut elimination theorem. I first introduce the cut-free sequent calculus, which has no cut rule. To strengthen the cut elimination a bit, I only allow the axiom rule for atomic formulae in the cut-free calculus. I denote the cut-free sequents as  $\Gamma \supset_c \Delta$  (I don't give the definition here; it is very similar to the definition **G**).

I give a semantic proof for cut elimination. This proof does not generalize easily to predicate calculus, the semantic proof for predicate calculus is way more involved. I also tried to give a syntactical proof, but failed due to time constraints.

The semantic proof goes as follows. First we prove soundness of the sequent calculus with cut, which states that the conclusion of any derivation (with cuts) is valid. The main part is to prove completeness of the cut-free calculus. Suppose that we are given a valid sequent. If no logical connectives occur in the sequent, then we prove that we can either apply the axiom rule or the falsum rule, using that the sequent is valid. On the other hand, if any connective occurs in the sequent, we apply the corresponding rule (in reverse) to eliminate that connective. We can then show the hypotheses to the rule are also valid, and they contain fewer connectives than the conclusion. So we can use a nonstructural induction to conclude that the hypotheses are provable, which finishes the proof.

We need the following definitions. The first line defines the notation for a valid sequent. Then we define the size of a formula and sequent, which in this case is the number of connectives occurring in it.

```

Definition Validates v Δ := ∃ A, In A Δ ∧ Is_true (TrueQ v A).
Notation "Γ ⊃ Δ" := (∀ v, Satisfies v Γ → Validates v Δ) (at level 80).
Fixpoint size A : nat := match A with
| # P ⇒ 0
| ⊥ ⇒ 0
| B ∨ C ⇒ S (size B + size C)
| B ∧ C ⇒ S (size B + size C)
| B → C ⇒ S (size B + size C)
end.
Definition sizeI := map_fold_right size plus 0.
Definition sizes Γ Δ := sizeI Γ + sizeI Δ.

```

This allows us to prove the following theorem by induction to  $n$ .

```
Theorem Gcf_complete_induction : ∀ n Γ Δ, sizes Γ Δ ≤ n → Γ =⊃ Δ → Γ ⊃c Δ.
```

Together with Soundness this proves cut elimination.

```
Theorem Cut_elimination : ∀ Γ Δ, Γ ⊃ Δ → Γ ⊃c Δ.
```

## 6 Conclusion and Future work

In this project I have shown that the proofs of basic but important theorems about propositional calculus can be implemented in a proof assistant relatively painlessly. Since many concepts in proof theory are naturally definable using induction, Coq's excellent support for inductive datatypes and proofs by induction makes the formalization of this subject relatively easy. The lack of strong automation in Coq does require that the user has to focus a lot on details in the proof.

This project can be seen as a proof of concept for formalizing theorems in proof theory. One can extend this project in numerous ways. One can consider nonclassical logics, and prove soundness and completeness results about them, like the completeness intuitionistic logic w.r.t. Kripke models. Another way to extend this is to consider predicate calculus instead of propositional calculus. This has been done in the literature already, for example to prove Gödel's first incompleteness theorem.<sup>[5]</sup> Another possible extension is to prove other theorems, like the Beth definability theorem or Craig's interpolation theorem.

## References

- [1] Matt Bishop. Foundations of constructive analysis. 1967.
- [2] Georges Gonthier. A computer-checked proof of the four colour theorem. 2005.
- [3] Georges Gonthier. Engineering Mathematics: The Odd Order Theorem Proof. *SIGPLAN Not.*, 48(1):1–2, January 2013.
- [4] The Coq development team. *The Coq proof assistant reference manual*. LogiCal Project, 2013. Version 8.4.
- [5] Russell O'Connor. Essential Incompleteness of Arithmetic Verified by Coq. 2005.

- [6] Vincent Siles and Hugo Herbelin. Pure Type System conversion is always typable. *Journal of Functional Programming*, 22(2):153 – 180, May 2012.
- [7] Floris van Doorn, Herman Geuvers, and Freek Wiedijk. Explicit Convertibility Proofs in Pure Type Systems. LFMTTP '13, pages 25–36. ACM, 2013.